

単一層拡散媒体に対する透過拡散実験 シミュレーション及び解析プログラム

(技術報告)

2001年1月

核燃料サイクル開発機構
東海事業所

本資料の全部または一部を複写・複製・転写する場合は、下記にお問い合わせください。

〒319-1184 茨城県那珂郡東海村大字村松4-49
核燃料サイクル開発機構
技術展開部 技術協力課

Inquiries about copyright and reproduction should be addressed to :
Technical Cooperation Section,
Technology Management Division,
Japan Nuclear Cycle Development Institute
4-49 Muramatsu, Tokai-mura, Naka-gun, Ibaraki, 319-1184
Japan

© 核燃料サイクル開発機構 (Japan Nuclear Cycle Development Institute)
2001

単一層拡散媒体に対する透過拡散実験シミュレーション 及び解析プログラム

(技術報告)

佐藤 治夫¹

要 旨

岩石などの単一層拡散媒体に対する透過拡散実験のシミュレーション及び解析のためのプログラム (TDROCK1.FOR) を開発した。プログラムは、科学技術計算に適している Pro-Fortran により作成し、解析法として比較的簡単な陽解差分法を用いた。解析では、これまでに取扱うことができなかったトレーサセル中の溶質濃度の時間変化を入力条件とすることができ、トレーサセルから測定セル側への溶質の拡散に伴うトレーサセル中での時間に対する濃度の減少、媒体空隙水中の濃度分布及び測定セル中の溶質濃度の経時変化などを計算することができる。また、入力条件として、両セル中の溶液体積や試料の直径及び厚さをパラメータとすることもできる。本プログラムは、既に拡散係数(見掛けの拡散係数、実効拡散係数)が求められているケースについて測定セル中の溶質濃度の経時変化について検証した結果、実測結果をよく説明することができた。このことから、本解析プログラムが実際の解析やシミュレーションに適用できることが確認された。本報では、透過拡散実験における理論的取扱い、解析のためのモデル、ソースプログラム例及びマニュアルについて説明する。

¹ 核燃料サイクル開発機構 東海事業所 環境保全・研究開発センター
処分研究部 放射化学研究グループ/処分バリア性能研究グループ

A Development of Simulation and Analytical Program for Through-diffusion Experiments for a Single Layer of Diffusion Media

Haruo SATO¹

ABSTRACT

A program (TDROCK1.FOR) for simulation and analysis of through-diffusion experiments for a single layer of diffusion media was developed. This program was made by Pro-Fortran language, which was suitable for scientific and technical calculations, and relatively easy explicit difference method was adopted for an analysis. In the analysis, solute concentration in the tracer cell as a function of time that we could not treat to date can be input and the decrease in the solute concentration as a function of time by diffusion from the tracer cell to the measurement cell, the solute concentration distribution in the porewater of diffusion media and the solute concentration in the measurement cell as a function of time can be calculated. In addition, solution volume in both cells and diameter and thickness of the diffusion media are also variable as an input condition. This simulation program could well explain measured result by simulating solute concentration in the measurement cell as a function of time for case which apparent and effective diffusion coefficients were already known. Based on this, the availability and applicability of this program to actual analysis and simulation were confirmed. This report describes the theoretical treatment for the through-diffusion experiments for a single layer of diffusion media, analytical model, an example of source program and the manual.

¹ Radiochemistry Group/Barrier Performance Group, Waste Isolation Research Division,
Waste Management and Fuel Cycle Research Center, Tokai Works,
Japan Nuclear Cycle Development Institute

目 次

1. 緒論	-----	1
2. 基本理論	-----	3
2.1 解析解	-----	3
2.2 数値解析による解法	-----	5
2.2.1 導関数の差分近似	-----	6
2.2.2 制御式	-----	8
2.2.3 微分方程式の離散化	-----	8
3. プログラムの特徴及びマニュアル	-----	12
3.1 プログラムの特徴	-----	12
3.2 入力パラメータと出力パラメータ	-----	13
3.3 データの入力と計算	-----	17
3.4 入出力例	-----	17
4. 解析プログラムの検証	-----	19
5. 結論	-----	20
6. 今後の課題	-----	20
7. 参考文献	-----	20
8. ソースプログラム例	-----	22

図及び表

図 1 透過拡散法による拡散実験概念の例	-----	2
図 2 $f(x)-x$ グラフによる差分近似の概念	-----	7
図 3 有限差分法における透過拡散実験解析の概念モデル	-----	9
図 4 泥岩中の Cs の透過拡散実験に対する測定セル中濃度の経時変化の 実測値(プロット)とシミュレーション結果の比較	-----	19
表 1 入力パラメータ	-----	15
表 2 出力パラメータ	-----	16
表 3 データ入力の FORMAT	-----	17
表 4 データの入出力例	-----	18

1. 緒 論

高レベル放射性廃棄物のガラス固化体より浸出した放射性核種は、その周囲を覆っている緩衝材（ベントナイトという粘土）中を拡散により移行するものと考えられている。これは、ベントナイトに対する透水係数が極めて小さいためであり、ほとんど動いていないに等しい（これをベントナイトの止水性という）。さらにこの緩衝材中を核種が拡散移行した後、岩体中を移行する。この岩体中の移行については、物質移行モデルの観点から、主として亀裂性媒体と多孔質媒体中の移行に分類される（動燃事業団, 1992; PNC, 1993）。亀裂性媒体では、岩体中に発達した亀裂中を主に移行経路とするが、亀裂から鉛直方向へのマトリックス拡散も同時に起こる。亀裂性媒体は、主として花崗岩のような固結している結晶質岩系の岩石によく見られる。このマトリックス部¹への拡散は、長期にわたって地層処分²の安全評価を行う場合には、その結果に大きな影響を及ぼす可能性が示唆されている。一方、多孔質媒体中の拡散では、どの方向にも連続的にかつ均一に空隙³が存在し、この空隙中を移行する。主に亀裂が発達していない砂岩、泥岩、凝灰岩などの堆積岩系の岩石によく見られる（Sato et al., 1997a）。これら一連の安全評価計算に必要なパラメータの1つに拡散係数が挙げられている。

安全評価上の入力データとして必要とされる拡散係数は実効拡散係数（以下 D_e と記す）であり、一般的には Through-diffusion（透過拡散）法（喜多ら, 1989; 熊田ら, 1990; Park et al., 1991; 佐藤ら, 1992a, b; 佐藤・渋谷, 1994; Kato et al., 1995; Sato et al., 1997a, b）による拡散実験から直接得ることができる。

図1に透過拡散法による拡散実験の概念を示す。実験に使用される拡散セルは、トレーサセル（高濃度側リザーバ）と測定セル（低濃度側リザーバ）より構成され、両セルは試料（この場合は岩石）を介して連結されている。トレーサセルにはトレーサ溶液が注入され、測定セルにはトレーサの含まれていない溶液が注入される。この結果、両セル間には試料を挟んで溶質の濃度勾配が形成される。この濃度勾配により高濃度側から低濃度側セルへ媒体空隙水³中を溶質が拡散し、低濃度側セル中の溶質濃度は時間と共に上昇する。一方、トレーサセル中の溶質濃度は逆に減少する。測定セル側での溶質濃度の時間に対する増加割合は、非定常の状態では一定でなく時間に対して非線形を為す。即ち、溶質のフラックスが時間と共に変化する。一方、非定常から定常状態になるにつれて溶質のフラックスは徐々に増加し、最終的には一定となり、濃度の時間に対する増加割合は一定となり、濃度は時間に対して直線的に増加する。この時間に対する濃度増加の傾きより D_e を求めることができる。また、定常状態になるまでのタイムラグ（時間遅れ）より分配係数（以下 K_d と記す）を求めることもできる。さらにこの両パラメータより見掛けの拡散係数（以下 D_a と記す）を求めることができる。このように、透過拡散法は非常に有利な測定手法であると言える。しかしながら、この測定方法には幾つかの制約もある。

¹ 岩石中の径の大きい粒の間隙や粉砕によって形成された割れ目の間隙が、微小な粒によって埋められている部分を言い、基質とも言う。

² 媒体の全体積の内、空間の占める割合。他に間隙率や孔隙率、あるいは空孔率（porosity）などとも呼ばれる。

³ 媒体の空隙を満たしている水を空隙水と言う。他に空隙率と同様、間隙水や孔隙水（pore water）などとも呼ばれる。

透過拡散実験を行う場合、拡散状態が非定常から定常状態になるまでの両セルの溶質の濃度変化を測定することになり、核種の試料に対する収着特性、溶液中での核種の溶解特性、試料の空隙率や空隙構造などの空隙特性などに依存し、実験期間が大きく左右される。分配係数（以下 K_d と記す）が大きい時は極めて長期に及ぶ場合もあり、実験期間を通して溶液条件を制御することは非現実的である。また、実験期間中に核種がセル内壁に吸着し、濃度が大きく変化してしまうこともあり、技術的に溶質濃度を一定に保つことが困難な場合もあり、現実的に実験を行うことができない核種も多く存在する。一般的に用いられている解析解は、トレーサセル側濃度と測定セル側濃度の差が大きく、測定セル側濃度がトレーサセル側濃度と比較して0と見なせる場合で、かつトレーサセル側濃度が一定の場合についてのみ成立する(Crank, 1975)。一般的にはこの条件を満たすように実験は行われているか、若しくは条件を満たせる元素に対して行われている。しかしながら、長期にわたる実験においては、比較的溶解度が高い可溶性の核種以外は溶液の pH の変化や岩石を構成する鉱物の溶解などによりトレーサセル側濃度が変化したり、核種の試料への収着やセル内壁への吸着が激しく、予想以上にトレーサセル側濃度が減少したりすることがしばしば起こる。これは即ち境界条件が変化することを意味するものであり、濃度勾配にも影響を与え、ひいてはフラックスに影響を与え、既存の境界濃度条件が一定の解析解によって単純に拡散係数を求めることはできない。このように、境界条件が時間によって変化した場合の解析法は課題の1つとなっている。

本報告では、両セル中の溶質濃度が任意の時間に变化した場合や、試料の直径や厚さなど、様々な試料条件に対して透過拡散実験を行った場合での試料空隙水中の溶質濃度分布や透過曲線などを数値解析的に計算し、フィッティングにより両拡散係数 (D_a , D_e) を求めるために開発した解析プログラムとそのマニュアルについて記述する。




図1 透過拡散法による拡散実験概念の例 (Sato et al., 1997b)

尚、本解析プログラムは、岩石などのように単一層拡散媒体を用いた場合を対象としており、ベントナイトのように、媒体の両側をフィルタで固定した場合の解析に対しては適用することができない。岩石の場合、拡散層が1層（岩石層）であるため、2種類の拡散係数（ De , Da ）を用いて解析するのに対して、ベントナイトの場合、ベントナイトを挟んでいるフィルタ中の拡散についても考慮しなければならないため、媒体中の拡散層が全部で3層（フィルタ2層、ベントナイト1層）となり、4種類の拡散係数（ベントナイト及びフィルタ中の De , Da ）が必要となること及び各層に分けて拡散を計算しなければならないこと、などから全く異なった計算となる。このような場合における解析については、今後検討しプログラムを開発する予定である。

2. 基本理論

2.1 解析解

一次元に対する単位面積を単位時間に通過する物質量は Fick の第 1 法則により以下の式で表される(Crank, 1975)。

$$J = -De \left(\frac{\partial C_p}{\partial X} \right) \quad (2.1)$$

ここで、

J : 単位面積を単位時間に通過する物質量 (フラックス) [$M/L^2/T$]

De : 実効拡散係数 [L^2/T]

C_p : 空隙水中の溶質濃度 [M/L^3]

X : 拡散源からの距離 [L]

空隙水中の溶質濃度 C_p と媒体中の溶質濃度 C との関係及びこの関係を用いて、(2.1)式の空隙水中の溶質濃度 C_p を媒体中の濃度 C で表せば次のように整理される。

$$C_p = \frac{C}{\varepsilon + \rho \cdot K_d} \quad (2.2)$$

$$J = -\frac{De}{\varepsilon + \rho \cdot K_d} \left(\frac{\partial C}{\partial X} \right) = -\frac{\varepsilon \cdot D_p}{\varepsilon + \rho \cdot K_d} \left(\frac{\partial C}{\partial X} \right) = -Da \left(\frac{\partial C}{\partial X} \right) \quad (2.3)$$

ここで、

Da : 見掛けの拡散係数 [L^2/T]

D_p : 空隙水中の拡散係数 [L^2/T]

C : 媒体中の溶質濃度 [M/L^3]

ε : 空隙率 [L^3/L^3]

ρ : 媒体の乾燥かさ密度 [M/L^3]

K_d : 分配係数 [L^3/M]

さらに、右辺第1項及び第2項の分母 $\varepsilon + \rho \cdot K_d (= \alpha)$ は収着容量 (rock capacity factor) と呼ばれ、媒体の溶質に対する収着性 (保持能力) の大きさを表わす。

媒体の乾燥かさ密度と真密度との関係は次式によって表わされる。

$$\rho = (1 - \varepsilon) \cdot \rho_{th} \quad (2.4)$$

ここで、

ρ_{th} : 媒体の真密度⁴ [M/L³]

ところで、媒体空隙水中の溶質の濃度分布は、Fick の第2法則に半減期を考慮して次のように表わされる。

$$\frac{\partial C_p}{\partial t} = \frac{\partial}{\partial X} \left(Da \frac{\partial C_p}{\partial X} \right) - \lambda \cdot C_p \quad (2.5)$$

ここで、 Da が場所によらず一定であるとすれば、(2.5)式は次のように整理される。

$$\frac{\partial C_p}{\partial t} = Da \left(\frac{\partial^2 C_p}{\partial X^2} \right) - \lambda \cdot C_p \quad (2.6)$$

ここで、

t : 拡散時間 [T]

λ : 崩壊定数 [1/T]

(2.6)式をさらに媒体中の濃度 C で表わせば、(2.2)式を用いて次のように整理される。

$$\frac{\partial C}{\partial t} = Da \left(\frac{\partial^2 C}{\partial X^2} \right) - \lambda \cdot C \quad (2.7)$$

半減期が実験期間と比較して十分に長い若しくは安定同位元素を用いた場合は、(2.5)、(2.6)及び(2.7)式の右辺第2項は近似的に無視することができるか、或いは考慮する必要は無くなる。

透過拡散実験において、トレーサセル側濃度が一定に保たれ、かつ測定セル側濃度が0と見なせる程小さいとすれば、任意の時刻での測定セル側濃度は、初期及び境界条件に基づいて以下のように導かれている(Crank, 1975)。

初期条件

$$C_p(X, t) = 0, t = 0, 0 \leq X \leq H$$

境界条件

⁴ 空隙率が0の場合の密度であり、固相密度 (solid density) とも呼ばれる。

$$C_p(X, t) = C_0, t > 0, X = 0$$

$$C_p(X, t) = 0, t > 0, X = H$$

$$C_d = \frac{S_r \cdot H \cdot C_0}{V_m} \left[\frac{De}{H^2} t - \frac{\alpha}{6} - \frac{2\alpha}{\pi^2} \times \sum_{n=1}^{\infty} \left\{ \frac{(-1)^n}{n^2} \cdot \exp\left(-\frac{De \cdot n^2 \cdot \pi^2 \cdot t}{H^2 \cdot \alpha}\right) \right\} \right] \quad (2.8)$$

ここで、

C_d : 測定セル中の溶質の濃度 [ML³]

C_0 : トレーサセル中の溶質濃度 [ML³]

S_r : サンプルの断面積 [L²]

H : サンプルの厚さ [L]

V_m : 両セル中の溶液体積 [L³]

α : rock capacity factor (収着容量) [-]

(2.8)式は、非定常の状態についても含んでいる解析解である。時間が経過し定常状態に達すると、濃度は時間に対して直線的に増加する。即ち(2.8)式の指数項の部分はほとんど無視し得ることを意味する。従って次のように整理される。

$$C_d = \frac{S_r \cdot H \cdot C_0}{V_m} \left(\frac{De}{H^2} t - \frac{\alpha}{6} \right) \quad (2.9)$$

De は、(2.9)式に基づいて定常状態時の時間に対する測定セル側溶質濃度の増加の傾きから求めることができる。また、次式によって与えられるタイムラグ (time-lag) 法 (Muurinen et al., 1987)により Da を求めることもできる。

$$\frac{De}{\alpha} = Da = \frac{H^2}{6T_{int}} \quad (2.10)$$

ここで、

T_{int} : (2.9)式を時間軸まで外挿したときの時間軸との交点 [T]

2.2 数値解析による解法

2.1 では理想的な条件で実験が行われた場合についての解析法について説明した。しかしながら、実際の実験ではこれらの解析解がそのまま適用できない場合もある。もしもトレーサセル側溶質濃度が一定ではなく時間とともに変化し、かつ濃度変化が関数形で表せない場合には解析解をそのまま適用することはできない。また、トレーサセル側溶質濃度を溶液の pH などを調整して溶解度制限にした場合などは、溶液の pH のずれなどにより濃度が変化する。即ち境界条件が変化する。このような場合は、境界条件が関数形ではないため解析解を導くことは困難である。

ここでは、有限差分法 (Finit Difference Method) の内、前進的に解くことができる

陽解法 (explicit method) による解析法について説明する。

2.2.1 導関数の差分近似

関数 $f(x)$ 及びその導関数が X の 1 価有限連続関数であるとき、Taylor の定理から

$$\begin{aligned}
 f(x + \Delta x) = f(x) + \Delta x \cdot f'(x) + \frac{1}{2} \Delta x^2 \cdot f''(x) + \frac{1}{6} \Delta x^3 \cdot f'''(x) + \dots \\
 \dots + \frac{1}{n!} \Delta x^n \cdot f^{(n)}(x) + \dots
 \end{aligned}
 \tag{2.11}$$

また、

$$\begin{aligned}
 f(x - \Delta x) = f(x) - \Delta x \cdot f'(x) + \frac{1}{2} \Delta x^2 \cdot f''(x) - \frac{1}{6} \Delta x^3 \cdot f'''(x) + \dots \\
 \dots + \frac{(-1)^n}{n!} \Delta x^n \cdot f^{(n)}(x) + \dots
 \end{aligned}
 \tag{2.12}$$

となる。(2.11)と(2.12)式を加えると、

$$f(x + \Delta x) + f(x - \Delta x) = 2f(x) + \Delta x^2 \cdot f''(x) + O(\Delta x^4)
 \tag{2.13}$$

を得る。ここで、 $O(\Delta x^4)$ は ΔX の 4 次以上を含む項を表すものとする。 $O(\Delta x^4)$ は ΔX のより低次の項と比較して無視できるほど小さいとすれば、(2.13)式は次のように整理される。

$$f''(x) = \left(\frac{d^2 f(x)}{d x^2} \right)_{x=x} \cong \frac{f(x + \Delta x) - 2f(x) + f(x - \Delta x)}{\Delta x^2}
 \tag{2.14}$$

この式の右辺の主要誤差は ΔX^2 の程度による。また、(2.11)と(2.12)式の差をとり、 ΔX^3 の項を無視すれば、

$$f'(x) = \left(\frac{d f(x)}{d x} \right)_{x=x} \cong \frac{f(x + \Delta x) - f(x - \Delta x)}{2\Delta x}
 \tag{2.15}$$

が導かれる。この式の主要誤差は ΔX の程度による。

図 2 に差分近似の概念を示す。(2.15)式は、点 P における接線の傾斜を弦 AB の傾斜で近似することであり、中心差分近似と呼ばれる。また、点 P における接線の傾斜は、弦 PB で近似する前進差分近似公式

$$f'(x) \cong \frac{f(x+\Delta x) - f(x)}{\Delta x} \quad (2.16)$$

あるいは弦 AP で近似する後退差分近似公式

$$f'(x) \cong \frac{f(x) - f(x-\Delta x)}{\Delta x} \quad (2.17)$$

で表すこともできる。(2.16)及び(2.17)式の主要誤差は、共に $O(\Delta x^2)$ であることが分かる。




図2 $f(x)-x$ グラフによる差分近似の概念

2.2.2 制御式 (governing equation)

単位時間に単位面積を通過する物質量を表わすフラックスの式は(2.1)式により与えられる。また、ある時刻での空隙水中の濃度分布は(2.6)式により与えられる。

トレーサセル側及び測定セル側での初期及び境界条件は以下のように与えられる。

初期条件

$$C_p(X, t) = 0, t = 0, 0 < X \leq H$$

$$C_p(X, t) = C_0, t = 0, X = 0$$

境界条件

(トレーサセル側)

$$\frac{\partial C_p}{\partial t} = -\frac{J \cdot S_r}{V_t} \Big|_{t>0, X=0} = \left(\frac{S_r}{V_t} D_e \right) \frac{\partial C_p}{\partial X} \Big|_{t>0, X=0}$$

(測定セル側)

$$\frac{\partial C_p}{\partial t} = \frac{J \cdot S_r}{V_m} \Big|_{t>0, X=H} = \left(-\frac{S_r}{V_m} D_e \right) \frac{\partial C_p}{\partial X} \Big|_{t>0, X=H}$$

ここで、


V_t : トレーサセル中の溶液体積 [L³]

V_m : 測定セル中の溶液体積 [L³]

2.2.3 微分方程式の離散化 (descretization)

図3に透過拡散実験に対する有限差分法(陽解法)による数値解析の概念モデルを示す。試料(岩石)のトレーサセル側の面を原点として、試料の深さ方向に等しく分割した時の距離の刻みを ΔX 、また時間の刻み幅を Δt とする。本解析プログラム

(TDROCK1.FOR)では、計算条件に応じて距離の分割数を5、10、15、20の中から選択できるようにプログラムされている。また、距離及び時間ステップ毎にそれぞれ番号を付している(距離:i, 時間:j)。各格子はある距離と時刻における空隙水中での溶質濃度を表わし、 $C_{pi, j}$ で表現する。



● 格子の影響領域(制御範囲) ● 有限差分格子

図3 有限差分法における透過拡散実験解析の概念モデル

(2.1)と(2.6)式、初期及び境界条件を陽解法による離散化式で表わせれば次のようになる。

$$C_p(X, t) = C_{p_i, j}$$

と定義すれば、(2.1)式におけるフラックスは、離散化式では次のように記述される。

($X=0$ でのフラックス)

$$J_{i, j} = -De \left(\frac{C_{p_{i+1}, j} - C_{p_{i, j}}}{\Delta X} \right) = De \left(\frac{C_{p_{i, j}} - C_{p_{i+1}, j}}{\Delta X} \right) \quad (2.18)$$

($i=1, j=1, 2, 3, \dots$)

($X=H$ でのフラックス)

$$J_{i, j} = -De \left(\frac{C_{p_{i, j}} - C_{p_{i-1}, j}}{\Delta X} \right) = De \left(\frac{C_{p_{i-1}, j} - C_{p_{i, j}}}{\Delta X} \right) \quad (2.19)$$

($i=n_x+1, j=1, 2, 3, \dots$)

($0 < X < H$ でのフラックス)

$$J_{i, j} = -De \left(\frac{C_{p_{i+1}, j} - C_{p_{i-1}, j}}{2\Delta X} \right) = De \left(\frac{C_{p_{i-1}, j} - C_{p_{i+1}, j}}{2\Delta X} \right) \quad (2.20)$$

($i=2, 3, 4, \dots, n_x, j=1, 2, 3, \dots$)

ここで、

n_x : 試料の厚さ方向における分割数 ($n_x = 5, 10, 15, 20$)

$J_{i, j}$: 距離及び時間ステップ (i, j) におけるフラックス

原点からの距離 X_i 及び時間 t_j は次の式から求められる。

$$X_i = (i-1) \cdot \Delta X \quad (i=1, 2, 3, \dots, n_x+1)$$

$$t_j = (j-1) \cdot \Delta t \quad (j=1, 2, 3, \dots)$$

(2.6)式の偏微分方程式について、半減期が無視できるとして、(2.14)及び(2.16)式に基づいて離散化式で表わせれば次のようになる。ただし、ここでは空隙水中の濃度で取扱った。

$$\begin{aligned}
\frac{C_{p_i, j+1} - C_{p_i, j}}{\Delta t} &= Da \left(\frac{C_{p_{i+1}, j} - C_{p_i, j}}{\Delta X} - \frac{C_{p_i, j} - C_{p_{i-1}, j}}{\Delta X} \right) / \Delta X \quad (2.21) \\
&= Da \left(\frac{C_{p_{i+1}, j} - 2C_{p_i, j} + C_{p_{i-1}, j}}{\Delta X^2} \right)
\end{aligned}$$

上式を整理して次式を得る。

$$\begin{aligned}
C_{p_i, j+1} &= C_{p_i, j} + Da \frac{\Delta t}{\Delta X^2} (C_{p_{i+1}, j} - 2C_{p_i, j} + C_{p_{i-1}, j}) \quad (2.22) \\
(i=2, 3, 4, \dots, n_x, j=1, 2, 3, \dots)
\end{aligned}$$

(2.22)式が数値分散を起こさないための条件は次式を満たすものである(藤原, 1971)。

$$Da \frac{\Delta t}{\Delta X^2} \leq \frac{1}{2} \quad (2.23)$$

これは即ち(2.22)式を

$$C_{p_i, j+1} = \left(1 - 2Da \frac{\Delta t}{\Delta X^2} \right) C_{p_i, j} + Da \frac{\Delta t}{\Delta X^2} (C_{p_{i+1}, j} + C_{p_{i-1}, j}) \quad (2.24)$$

と書き換えると、 $Da \cdot \Delta t / \Delta X^2 > 0$ であり、 $C_{p_{i+1}, j}$ 、 $C_{p_{i-1}, j}$ 、 $C_{p_i, j}$ も正であるから、 $C_{p_i, j+1}$ が正であるためには $1 - 2Da \cdot \Delta t / \Delta X^2 \geq 0$ でなければならないことによる。

初期条件は次のように表わされる。

$$C_{p_i, j} = C_0, \quad i=1, j=1$$

$$C_{p_i, j} = 0, \quad i=2, 3, \dots, n_x + 1, j=1$$

また、境界条件は次のように表わされる。

(X=0側)

$$\frac{C_{p_i, j+1} - C_{p_i, j}}{\Delta t} = -\frac{J_{i, j} \cdot Sr}{V_t} = \frac{Sr}{V_t} De \left(\frac{C_{p_{i+1}, j} - C_{p_i, j}}{\Delta X} \right) \quad (2.25)$$

(i=1, j=1, 2, 3, ……)

(X=H側)

$$\frac{C_{p_i, j+1} - C_{p_i, j}}{\Delta t} = \frac{J_{i, j} \cdot Sr}{V_m} = \frac{Sr}{V_m} De \left(\frac{C_{p_{i-1}, j} - C_{p_i, j}}{\Delta X} \right) \quad (2.26)$$

($i = n_x + 1, j = 1, 2, 3, \dots$)

よって、(2.25)と(2.26)式はそれぞれ次のように整理される。

$$\begin{aligned} C_{p_i, j+1} &= C_{p_i, j} - \Delta t \frac{J_{i, j} \cdot Sr}{V_t} \\ &= C_{p_i, j} - \Delta t \frac{Sr}{V_t} De \left(\frac{C_{p_i, j} - C_{p_{i+1}, j}}{\Delta X} \right) \end{aligned} \quad (2.27)$$

($i = 1, j = 1, 2, 3, \dots$)

$$\begin{aligned} C_{p_i, j+1} &= C_{p_i, j} + \Delta t \frac{J_{i, j} \cdot Sr}{V_m} \\ &= C_{p_i, j} + \Delta t \frac{Sr}{V_m} De \left(\frac{C_{p_{i-1}, j} - C_{p_i, j}}{\Delta X} \right) \end{aligned} \quad (2.28)$$

($i = n_x + 1, j = 1, 2, 3, \dots$)

これらの差分式中で、トレーサセル中の濃度の経時変化に関する情報（実測された濃度と時刻）は、(2.18)式中の境界濃度 $C_{p_{i,j}}$ ($X = 0$ ($i = 1$) での濃度) へ入力される。また、 $X = 0$ ($i = 1$) 及び $X = H$ ($i = n_x + 1$) における濃度 $C_{p_{i,j}}$ と $C_{p_{n_x+1,j}}$ は、それぞれトレーサセル側及び測定セル側の濃度を表わす。

3. プログラムの特徴及びマニュアル

3.1 プログラムの特徴

プログラムは Fortran (Pro-Fortran) にて作成しており、データの入力は対話形式ではなく、データ入力ファイル (TDROCK1.DAT) へ入力する形式を採用している。以下にプログラムの特徴を列挙する。

- (1) 入力データや出力データを自動的にプリンターへプリントアウトするか否か選択でき、プリントアウトする場合は、プリントアウト回数及び時間（拡散実験での経過時間）を最大 20 回まで設定できる。
- (2) 拡散セルの内、各セルの容量（溶液体積）を個別に設定できる。
- (3) 実験に使用した試料の厚さと直径を自由に設定でき、厚さ方向の差分における分割数も 5, 10, 15, 20 の中から数値分散しない条件となるように選択できる。分割数はこれ以外の数を入力すると計算せず、警告を表示して計算を終了する。また、試料が円盤状ではない時は、等価直径で入力することができる。
- (4) トレーサセル側溶液の液を全量途中で交換する場合、交換する回数、交換する時間、及び交換した溶質濃度を最大 20 回まで自由に設定できる。

- (5)計算が発散しないような入力になっているかを判定する機能が施されており、発散する条件が入力された場合は、計算する前に判定結果と警告を表示するとともに計算を終了する。これは、 $Da \cdot \Delta t / \Delta X^2$ を入力された値で計算し、発散条件の 0.5 を超えると発散警告を出すようになっているためである。
- (6)溶質濃度やフラックスの単位についても、ホットとコールド実験の両方に対応できるように 4 種類 (濃度: Bq/ml, ppm, ppb, M; フラックス: Bq/cm²/s, μg/cm²/s, ng/cm²/s, mM/cm²/s) から選択することができ、計算もこれに対応して行われ、アウトプットも選択した単位に応じて表示される。
- (7)時間の単位も入力時については時間 (hour) のみであるものの、出力時には時間 (hour) と日 (day) の両方で表示される。
- (8)試料空隙水中溶質濃度分布の計算結果は、比較のため、境界条件一定 (トレーサセル側濃度 C_0 で一定と測定セル側濃度 0 で一定) の場合の計算結果も同時に出力される。

3.2 入力パラメータと出力パラメータ

表 1 に入力パラメータ、表 2 に出力パラメータを示す。同時に各パラメータの意味及び入力制限についても同表において説明する。

入力パラメータには以下に示す 19 種類がある。

- (1)プリンターへのプリントアウトの選択数
- (2)トレーサセル中の初期トレーサの濃度
- (3)測定セル中の初期のトレーサ濃度
- (4)濃度単位の選択数
- (5)見掛けの拡散係数
- (6)実効拡散係数
- (7)サンプルの厚さ
- (8)サンプルの厚さ方向の分割数
- (9)計算時間の刻み
- (10)サンプルの直径
- (11)トレーサセル中の溶液の体積
- (12)測定セル中の溶液の体積
- (13)トレーサセル中のトレーサ溶液の交換選択数
- (14)トレーサを交換する回数
- (15)トレーサセル中のトレーサの交換時刻
- (16)交換されたトレーサセル中のトレーサ濃度
- (17)計算したい最終時間 (拡散実検期間が望ましい)
- (18)計算結果をプリントアウトする回数
- (19)計算結果をプリンターへプリントアウトしてほしい時刻

また、出力パラメータには以下に示す 20 種類がある。

- (1)プリンターへのプリントアウトの選択数

- (2)トレーサセル中の初期トレーサの濃度
- (3)測定セル中の初期のトレーサ濃度
- (4)濃度単位の選択数
- (5)見掛けの拡散係数
- (6)実効拡散係数
- (7)サンプルの厚さ
- (8)サンプルの厚さ方向の分割数
- (9)計算時間の刻み
- (10)サンプルの直径
- (11)トレーサセル中の溶液の体積
- (12)測定セル中の溶液の体積
- (13)トレーサセル中のトレーサ溶液の交換選択数
- (14)トレーサを交換する回数
- (15)計算したい最終時間
- (16)計算結果をプリントアウトする回数
- (17)各時刻に対するトレーサセル側入口と測定セル側出口でのフラックス
- (18)トレーサセル側試料面から各距離の刻みごとの空隙水中の濃度（濃度分布）
- (19)境界条件一定の場合の各時刻に対するトレーサセル側入口と測定セル側出口でのフラックス
- (20)境界条件一定の場合のトレーサセル側試料面から各距離の刻みごとの空隙水中の濃度（濃度分布）

表1 入力パラメータ

No	記号	パラメータ	単位	意味	備考
1	P	プリンターへのプリントアウトの選択数	整数	P=1: プリンターへ自動的にプリントアウト P=0: プリントアウトしない	P=0 または 1 以外の数字を入力すると STOP すると共に警報が表示される 出力時の記号: P
2	CH	トレーサセル中の初期トレーサ濃度	Bq/ml ppm ppb M	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	NU: 濃度単位選択数 出力時の記号: Ch
3	CL	測定セル中の初期トレーサ濃度	Bq/ml ppm ppb M	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	NU: 濃度単位選択数 出力時の記号: Cl
4	NU	濃度単位選択数	整数	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	出力時の記号: NU
5	DA	見掛けの拡散係数	cm ² /s		出力時の記号: Da
6	DE	実効拡散係数	cm ² /s		出力時の記号: De
7	Z	試料の厚さ	cm		出力時の記号: Z
8	NZ	試料の厚さ方向の分割数	整数	NZ=5, 10, 15, 20 の中から選択	NZ=5, 10, 15, 20 以外の数字を入力すると STOP すると共に警告が表示される。 出力時の記号: Nz
9	DT	計算時間の刻み	hr		出力時の記号: DT
10	D	試料の直径	cm		出力時の記号: D
11	VT	トレーサセル中の溶液体積	ml		出力時の記号: Vt
12	VM	測定セル中の溶液体積	ml		出力時の記号: Vm
13	ND	トレーサセル中のトレーサ溶液の交換選択数	整数	ND=1: 交換する ND≠1: 交換しない	出力時の記号: ND
14	NT	トレーサ溶液を交換する回数	整数	1 ≤ NT ≤ 20: 最大 20 回まで交換可能	ND=1 の時機能する 出力時の記号: NT
15	TT	トレーサセル中のトレーサ溶液の交換時刻	hr	トレーサを交換(濃度が変化)した時刻を順に入力(最大 20 回まで)	ND=1 の時機能する 出力時の記号: 出力されない
16	CHC	交換(濃度が変化)されたトレーサセル中のトレーサ濃度	Bq/ml ppm ppb M	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M トレーサを交換(濃度が変化)した時刻毎のトレーサセル中のトレーサ濃度を順に入力	ND=1 の時機能する 出力時の記号: 出力されない
17	TC	計算したい最終時間(基本的には拡散期間が望ましい)	hr	入力した時刻まで計算した時点で計算を STOP する	出力時の記号: Tc, Tcd 出力は hr(Tc) と day(Tcd) の両方を表示
18	NTP	計算結果をプリンターへプリントアウトする回数	整数	1 ≤ NTP ≤ 20: 最大 20 回まで出力可能	NTP=0 を入力すると、出力結果は画面上のみの出力となる 出力時の記号: NTP
19	TP	計算結果をプリンターへプリントアウトしてほしい時刻	hr	最大 20 回まで入力可能 プリンターへ出力したい時刻を順に記入	P=1, 1 ≤ NTP ≤ 20 の時機能 出力時の記号: 出力されない

表2 出力パラメータ

No	記号	パラメータ	単位	意味	備考
1	P	プリンターへのプリントアウトの選択数	整数	P=1: プリンターへ自動的にプリントアウト P=0: プリントアウトしない	P=0または1以外の数字を入力すると STOP すると共に警報が表示される 出力時の記号: P
2	CH	トレーサセル中の初期トレーサ濃度	Bq/ml ppm ppb M	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	NU: 濃度単位選択数 出力時の記号: Ch
3	CL	測定セル中の初期トレーサ濃度	Bq/ml ppm ppb M	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	NU: 濃度単位選択数 出力時の記号: Cl
4	NU	濃度単位選択数	整数	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	出力時の記号: NU
5	DA	見掛けの拡散係数	cm ² /s		出力時の記号: Da
6	DE	実効拡散係数	cm ² /s		出力時の記号: De
7	Z	試料の厚さ	cm		出力時の記号: Z
8	NZ	試料の厚さ方向の分割数	整数	NZ=5, 10, 15, 20の中から選択	NZ=5, 10, 15, 20以外の数字を入力すると STOP すると共に警告が表示される。 出力時の記号: NZ
9	DT	計算時間の刻み	hr		出力時の記号: DT
10	D	試料の直径	cm		出力時の記号: D
11	VT	トレーサセル中の溶液体積	ml		出力時の記号: Vt
12	VM	測定セル中の溶液体積	ml		出力時の記号: Vm
13	ND	トレーサセル中のトレーサ溶液の交換選択数	整数	ND=1: 交換する ND≠1: 交換しない	出力時の記号: ND
14	NT	トレーサ溶液を交換する回数	整数	1≤NT≤20: 最大20回まで交換可能	ND=1の時機能する 出力時の記号: NT
15	Tc	計算したい最終時間	hr	入力した時刻まで計算した時点で計算をSTOPする	出力時の記号: Tc, Tcd 出力はhr(Tc)とday(Tcd)の両方を表示
16	NTP	計算結果をプリンターへプリントアウトする回数	整数	1≤NT≤20: 最大20回まで出力可能	NTP=0を入力すると、出力結果は画面の上のみの出力となる 出力時の記号: NTP
17	FLAX1 FLAX2	各時刻に対するトレーサセル側入口(FLAX1)と出口(FLAX2)でのフラックス	Bq/cm ² /s μg/cm ² /s ng/cm ² /s mM/cm ² /s	NU=1: Bq/cm ² /s NU=2: μg/cm ² /s NU=3: ng/cm ² /s NU=4: mM/cm ² /s	NU: 濃度単位選択数 出力時の記号: FLAX1, FLAX2
18	C	トレーサセル側試料面から各距離の刻み毎の空隙水中の濃度(濃度分布)	Bq/ml ppm ppb M	NU=1: Bq/ml NU=2: ppm NU=3: ppb NU=4: M	NU: 濃度単位選択数 出力時の記号: C
19	FFLAX1 FFLAX2	境界条件一定の場合の各時刻に対するトレーサセル側入口側(FFLAX1)と出口側(FFLAX2)でのフラックス	Bq/cm ² /s μg/cm ² /s ng/cm ² /s mM/cm ² /s	NU=1: Bq/cm ² /s NU=2: μg/cm ² /s NU=3: ng/cm ² /s NU=4: mM/cm ² /s	NU: 濃度単位選択数 出力時の記号: FFLAX1, FFLAX2

20	CC	境界条件一定の場合のトレーサセル側試料面から各距離の刻み毎の空隙水中の濃度(濃度分布)	Bq/ml ppm ppb M	NU=1 : Bq/ml NU=2 : ppm NU=3 : ppb NU=4 : M	NU : 濃度単位選択数 出力時の記号 : CC
----	----	---	--------------------------	--	-----------------------------

3.3 データの入力と計算

データの入力は、データファイル (TDROCK1. DAT) の中に表 3 に示す Format で作成する。Format 通り作成したら、ファイルを save して閉じる。次に Program :TDROCK1. FOR を run させれば計算は実行される。

表 3 データ入力の FORMAT

File name: TDROCK1. DAT

P : プリンターへのプリントアウトの選択数(整数)
CH : トレーサセル中の初期のトレーサ濃度(実数値)
CL : 測定セル中の初期のトレーサ濃度(実数値)
NU : 濃度単位選択数(1 or 2 or 3 or 4)
DA : 見掛けの拡散係数(実数値)
DE : 実効拡散係数(実数値)
Z : 試料の厚さ(実数値)
NZ : 試料の厚さ方向の分割数(5 or 10 or 15 or 20)
DT : 計算時間の刻み(実数値)
D : 試料の直径(実数値)
VT : トレーサセル中の溶液の体積(実数値)
VM : 測定セル中の溶液の体積(実数値)
ND : トレーサセル中のトレーサ溶液の交換選択数(1 or 1 以外)
NT : トレーサ溶液を交換(濃度変化)する回数(整数)
TT : トレーサセル中のトレーサ溶液交換(濃度変化)時刻(TT(1) TT(2) TT(3) ……)
CHC : 交換(濃度変化)されたトレーサセル中のトレーサ濃度(実数値)
TC : 計算したい最終時刻(実数値)
NTP : 計算結果をプリンターへプリントアウトする回数(整数)
TP : 計算結果をプリンターへプリントアウトしてほしい時刻(TP(1) TP(2) TP(3) TP(4) ……)

3.4 入出力例

表 4 に実際の入力 (入力データファイル例) 及び出力例 (出力データのアウトプット例) を示す。計算例は、泥岩に対して行った Cs の透過拡散実験の解析ケース(佐藤, 1997, 1998)である。Program を実際に run させ、動作することを確認している。

表4 データの入出力例

データの入力例(入力データファイル)(1/2)

```

1
12000.0
0.0
2
3.9E-9
4.8E-9
0.5
20
10.0
3.0
100.0
100.0
0
1
0.0
12000.0
12000.0
18
0.0 480.0 1200.0 1920.0 2160.0 2400.0 2640.0 2880.0 3120.0
3600.0 4080.0 4800.0 6000.0 7200.0 8400.0 9600.0 10800.0
12000.0

```

データの入出力例(出力データのアウトプット)(2/2)

```

P= リンターへのシュツリヨクハンテイ P=1 スル
シヨキノコウノウト`カ`ワノウト` Ch= 12000.00 ppm
シヨキノテイノウト`カ`ワノウト` Cl= 0.00 ppm
ノウト`ノタンイハンテイキコ`ウ NU=2 タンイ:ppm
ミカケノカクサンケイスウ Da= 0.390E-08 cm2/s
シ`ッコウカクサンケイスウ De= 0.480E-08 cm2/s
カンセキノアツサ Z= 0.50 cm
アツサホウコウノファンカツスウ Nz=20 ファンカツ
シ`カンノキサ`ミ DT= 10.00 hr
サンブ`ルノチヨツケイ D= 3.00 cm
トレ`サーセルチュウノヨウエキノタイセキ Vt= 100.0 ml
ソクテイセルチュウノヨウエキノタイセキ Vm= 100.0 ml
トレ`サーノコウカンノハンテイスウ ND=0 コウカンシナイ
トレ`サーノコウカンカイスウ NT= 1 カイ
ケイサンシタイシ`カン Tc= 12000.00 hr
ケイサンシタイニスウ Tcd= 500.00 day
ケイサンケツカラア`リントアウトスルカイスウ NTP=18 カイ

ケイカシ`カン T= 0.00 hr
ケイカニスウ Td= 0.00 day

X=0デ`ノフラックス(キョウカishi`ヨウケン`エンカ) FLAX1= 0.230E-02 ug/cm2/s
X=Zデ`ノフラックス(キョウカishi`ヨウケン`エンカ) FLAX2= 0.000E+00 ug/cm2/s

ノウト` (キョウカishi`ヨウケン`エンカ) C= 12000.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`エンカ) C= 0.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`エンカ) C= 0.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`エンカ) C= 0.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`エンカ) C= 0.00 0.00 0.00 0.00 0.00 ppm

X=0デ`ノフラックス(キョウカishi`ヨウケン`イツテイ) FFLAX1= 0.230E-02 ug/cm2/s
X=Zデ`ノフラックス(キョウカishi`ヨウケン`イツテイ) FFLAX2= 0.000E+00 ug/cm2/s

ノウト` (キョウカishi`ヨウケン`イツテイ) CC= 12000.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`イツテイ) CC= 0.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`イツテイ) CC= 0.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`イツテイ) CC= 0.00 0.00 0.00 0.00 0.00 ppm
ノウト` (キョウカishi`ヨウケン`イツテイ) CC= 0.00 0.00 0.00 0.00 0.00 ppm

```

4. 解析プログラムの検証

本解析プログラムの理論的検証と適用性について検討した。既に見掛けの拡散係数と実効拡散係数が実測値から直接解析解で求められているケースについて、測定セル中のトレーサの濃度の経時変化をシミュレーションし実験結果と比較した。検証例として、3.4で入出力例として示した実験例(佐藤, 1997, 1998)を用いた。入出力データは表4に示す通りである。図4に測定セル中のCs濃度の経時変化の計算結果を実測結果とともに示す。シミュレーションの結果は、実測のプロット(○印)とよく一致していることが分かる。このことから、本解析プログラムは、実際の解析やシミュレーションに適用できることが確認された。




図4 泥岩中のCsの透過拡散実験に対する測定セル中濃度の経時変化の実測値(プロット)とシミュレーション結果の比較(実測値: 佐藤, 1997, 1998)

5. 結論

岩石などの単一層拡散媒体に対して透過拡散実験を行った場合の解析及びシミュレーションのための理論的取扱い、並びにモデルを検討し、プログラム (TDROCK1.FOR) を陽解差分法に基づいて開発した。プログラム化は Pro-Fortran により行い、マニュアルを作成した。解析においては、トレーサセル中の溶質濃度の経時変化を入力条件とすることができ、トレーサセルから測定セル側への溶質の拡散に伴うトレーサセル中での時間に対する濃度の減少、媒体空隙水中の濃度分布及び測定セル中の溶質濃度の経時変化などを計算することができる。さらに、入力条件として、両セル中の溶液体積や試料の直径及び厚さをパラメータとすることもできる。本プログラムは、既に拡散係数 (D_a , D_e) が求められているケースについて測定セル中の溶質濃度の経時変化を検証した結果、実測結果をよく説明することができた。このことから、本プログラムが実際の解析やシミュレーションに適用できることが確認された。

6. 今後の課題

本解析プログラムは岩石などのように硬く、試料の両側に支持を必要しない、単一層での拡散媒体を対象としており、ベントナイトについては現状では適用できない。ベントナイトを取扱った場合は、ベントナイトの両側に膨潤により膨らむことを抑えるため、フィルタを配置している。このような場合は、両側のフィルタ中の拡散についても考慮しなければならず、複数層（この場合は、フィルタ層、ベントナイト層、フィルタ層の3層）の媒体中の拡散を取扱う必要がある。また、本モデルにおいては、短半減期の放射性元素を用いた場合に対応するための半減期補正が考慮されていない。これらの問題を考慮した解析プログラムについては今後開発を進める予定である。

7. 参考文献

- (1) 喜多治之、岩井孝幸、中嶋悟：花崗岩および凝灰岩間隙水中のイオンの拡散係数の測定、応用地質、30巻2号、pp.26-32 (1989).
- (2) 熊田政弘、岩井孝幸、鈴木隆次、西山勝栄：岩石中のイオンの拡散と空隙の性質、鉱山地質、40(5)、pp.323-336 (1990).
- (3) 佐藤治夫、芦田敬、小原幸利、油井三和、梅木博之、石黒勝彦：ベントナイト及び岩石中の核種の実効拡散係数、動燃技術資料、PNC TN8410 92-164 (1992a).
- (4) 佐藤治夫、油井三和、石川博久：岩石中の空隙構造に関する研究、動燃技術資料、PNC TN8410 92-222 (1992b).
- (5) 佐藤治夫、渋谷朝紀：緩衝材及び岩石中への核種の吸着・拡散メカニズム、動燃技報 No.91、PNC TN8410 94-284、pp.71-89 (1994).
- (6) 佐藤治夫：放射性廃棄物地層処分における緩衝材および岩石中の核種移行に関する研究、秋田大学博士論文、p.173 (1997).
- (7) 佐藤治夫：放射性廃棄物地層処分における緩衝材および岩石中の核種移行に関する研究、動燃技術資料、PNC TN1410 97-202、p.174 (1998).
- (8) 藤川洋一郎 訳 (G. D. Sumith 著)：電算機による偏微分方程式の解法、(株)サイエンス社、p.17 (1971).

- (9) 動燃事業団：高レベル放射性廃棄物地層処分研究開発の技術報告書－平成3年度－、動燃技術資料、PNC TN1410 92-081 (1992).
- (10) J. Crank : The Mathematics of Diffusion, 2nd edn., Oxford University Press, London (1975).
- (11) H. Kato, M. Muroi, N. Yamada, H. Ishida, H. Sato : Estimation of Effective Diffusivity in Compacted Bentonite, Scientific Basis for Nuclear Waste Management XVIII, Vol.353, pp.277-284 (1995).
- (12) A. Muurinen, P. Penttilä-Hiltunen, J. Rantanen : Diffusion Mechanisms of Strontium and Cesium in Compacted Sodium Bentonite, Scientific Basis for Nuclear Waste Management X, Vol.84, pp.803-811 (1987).
- (13) C. K. Park, K. W. Han, H. H. Park : Determination of Diffusivities of Sorbing and Non-sorbing Species in Rocks, Proceedings of the Second Annual International Conference on the High Level Radioactive Waste Management, Vol.1, pp.156-160 (1991).
- (14) Power Reactor and Nuclear Fuel Development Corporation (PNC) : Research and Development on Geological Disposal of High Level Radioactive Waste –First Progress Report–, PNC TN1410 93-059 (1993).
- (15) H. Sato, T. Shibutani, M. Yui : Experimental and Modelling Studies on Diffusion of Cs, Ni and Sm in Granodiorite, Basalt and Mudstone, Journal of Contaminant Hydrology 26, Elsevier Science B. V., Netherland, pp.119-133 (1997a).
- (16) H. Sato, T. Shibutani, Y. Tachi, K. Ota, K. Amano, M. Yui : Diffusion Behaviour of Nuclides Considering Pathways in Fractured Crystalline Rocks, PNC TN8410 97-127 (1997b).

8. ソースプログラム例

```

type B:TDROCK1.FOR
                                PROGRAM TDROCK1
C*****
C*****
C
C   Calculations of diffusion in rock and change in
C   concentration of tracer in solution with time
C   for a through-diffusion experiment
C   (スルーディフュージョンホウニオケル カンセキチュウノ カクサンオヨビヨウエキチュウノ
C   トレーサーノウトノケイシヘンカノ ケイサンプログラム)
C*****
C
C                               Created by Haruo Sato
C
C                               Created April 10, 1998
C                               Revised January 5, 2001
C*****
C*****
C
C   P:symbol for judgement of printing(プリントアウトノハンテイ
C   キコウ)(P=1:print,P=0:not print)
C
C   CH:initial concentration of tracer in the tracer
C   cell(トレーサーセルチュウノシヨキノトレーサーノウト)(Ch)
C   (Bq/ml or ppm or ppb or M)(if NU=1, unit is
C   Bq/ml, if NU=2, unit is ppm, if NU=3, unit is
C   ppb, if UN=4, unit is M)
C
C   CL:initial concentration of tracer in the
C   measurement cell(ソクテイセルチュウノシヨキノトレーサーノウト)
C   (Cl)(Bq/ml or ppm or ppb or M)(if NU=1, unit is
C   Bq/ml, if NU=2, unit is ppm, if NU=3, unit is
C   ppb, if NU=4, unit is M)
C
C   NU:symbol for judgement of unit for concentration
C   (NU=1:Bq/ml, NU=2:ppm, NU=3:ppb, NU=4:M)
C   (ノウトノタンイハンテイキコウ)
C
C   C:concentration profile of tracer in the porewater
C   of rock(カンセキノクウケキスイチュウノトレーサーノウトフンブ
C   (キョウカイシヨウケンガヘンカスルハアイ))(NU=1:Bq/ml, NU=2:ppm,
C   NU=3:ppb, NU=4:M)
C
C   CC:concentration profile of tracer in the porewater
C   of rock(カンセキノクウケキスイチュウノトレーサーノウトフンブ
C   (キョウカイシヨウケンイツテイノハアイ))(NU=1:Bq/ml, NU=2:ppm,
C   NU=3:ppb, NU=4:M)

```

C
 C DA:apparent diffusion coefficient(ミカケノカクサンケイスウ)(Da)
 C (cm²/s)
 C
 C DE:effective diffusion coefficient(シッコウカクサンケイスウ)
 C (De)(cm²/s)
 C
 C Z:thickness of rock(ガンセキノアツサ)(cm)
 C
 C NZ:the number of divisions in the direction of
 C thickness(アツサホウコウノフンカツスウ)(Nz)(Nz:5 or 10 or 15
 C or 20)
 C
 C DT:width of time for simulation
 C (シミュレーションニオケルシカンノキサミ)(hr)
 C
 C T:elapsed time(ケイカシカン)(T)(hr)
 C
 C TD:elapsed time(ケイカニツスウ)(Td)(day)
 C
 C D:diameter of rock(ガンセキノチヨウケイ)(D)(cm)
 C
 C VT:solution volume in the tracer cell
 C (トレサーセルチュウノヨウエキノタイセキ)(Vt)(ml)
 C
 C VM:solution volume in the measurement cell
 C (ソクテイセルチュウノヨウエキノタイセキ)(Vm)(ml)
 C
 C ND:symbol for judgement for exchange of tracer in
 C the tracer cell(if ND=1, tracer is exchanged,
 C but if ND is not 1, it isn't exchanged)
 C (トレサーセルチュウノトレサーヨウエキノコウカンハンテイキコウ:ND=1ナラハ、
 C トレサーハコウカンサレ、ソレイガイハコウカンサレナイ)
 C
 C NT:the number of time exchanged tracer(max:20)
 C (トレサーヲコウカンシタカイスウ(サイタ`イ20カイ))
 C
 C TT:exchange time of tracer(トレサーノコウカンシカン)(hr)
 C
 C CHC:concentration of tracer exchanged in the tracer
 C cell(トレサーセルチュウデ`ノコウカンサレタトレサーノノウト`)(Bq/ml or
 C ppm or ppb or M)
 C
 C TC:final time to be calculated(ケイサンシタイシカン)(Tc)(hr)
 C
 C TCD:the number of day to be calculated(ケイサンシタイニツスウ)
 C (Tcd)(day)
 C
 C NTP:the number of time to be printed out for
 C calculated results(max:20)
 C (ケイサンシタケツカヲフ°リントアウトスルカイスウ)(サイタ`イ20カイ)
 C
 C TP:time to be printed out for calculated results
 C (ケイサンシタケツカヲフ°リントアウトスルシカン)(Tp)(hr)
 C

```

C FLAX1:diffusional mass flux at X=0
C      (X=0テ`ノフラックス(キョウカイシ`ヨウケンカ`ヘンカスルハ`アイ))(Bq/cm2/s
C      or ug/cm2/s or ng/cm2/s or mM/cm2/s)
C
C FLAX2:diffusional mass flux at X=Z(X=Zテ`ノフラックス
C      (キョウカイシ`ヨウケンカ`ヘンカスルハ`アイ))(Bq/cm2/s or ug/cm2/s
C      or ng/cm2/s or mM/cm2/s)
C
C FFLAX1:diffusional mass flux at X=0(X=0テ`ノフラックス
C      (キョウカイシ`ヨウケンカ`イッテイノハ`アイ))(Bq/cm2/s or ug/cm2/s
C      or ng/cm2/s or mM/cm2/s)
C
C FFLAX2:diffusional mass flux at X=Z(X=Zテ`ノフラックス
C      (キョウカイシ`ヨウケンカ`イッテイノハ`アイ))(Bq/cm2/s or ug/cm2/s
C      or ng/cm2/s or mM/cm2/s)
C
C*****
C
C      Declarations of Dimension and Valuables and
C      Data file open
C      (シ`ケンオヨヒ`ヘンスウノカタノセンケンオヨヒ`テ`-タファイルノカイシ`)
C
C*****
C
C      INTEGER P, NU, NZ, ND, NT, NTP
C      REAL CH, CL, C, CC, DA, DE, Z, DT, T, TD, D, VT,
C      &      VM, TT, CHC, TC, TCD, TP, FLAX1, FLAX2,
C      &      FFLAX1, FFLAX2, DX, A, DD, ZN
C
C      DIMENSION C(21,251), CC(21,251), TT(20), CHC(20),
C      &      TP(20)
C
C      OPEN(1,FILE='B:HARU1.DAT')
C      OPEN(7,FILE='PRN',STATUS='NEW')
C
C*****
C
C      Input of Data
C      (テ`-タノニューリョク)
C
C*****
C      Reading of data(テ`-タノヨミコミ)
C*****
C
C      READ(1,*) P
C      READ(1,*) CH
C      READ(1,*) CL
C      READ(1,*) NU
C      READ(1,*) DA
C      READ(1,*) DE
C      READ(1,*) Z
C      READ(1,*) NZ
C
C      IF(NZ.EQ.5.OR.NZ.EQ.10.OR.NZ.EQ.15.OR.NZ.EQ.20)
C      &      THEN
C      GO TO 1020
C      ELSE

```

```

WRITE(6,1010)
GO TO 500
ENDIF
C
1020 READ(1,*) DT
READ(1,*) D
READ(1,*) VT
READ(1,*) VM
READ(1,*) ND
READ(1,*) NT
C
IF(NT.GE.1.AND.NT.LE.20) THEN
READ(1,*) (TT(I),I=1,NT)
READ(1,*) (CHC(I),I=1,NT)
ELSE
WRITE(6,177)
GO TO 500
ENDIF
C
READ(1,*) TC
READ(1,*) NTP
C
IF(NTP.GE.1.AND.NTP.LE.20) THEN
READ(1,*) (TP(K),K=1,NTP)
ENDIF
C
C*****
C      Setting of parameter (ハ°ラメ-ター-ノセツテイ)
C*****
C
PAI=3.141592
ZN=FLOAT(NZ)
DX=Z/ZN
A=(D/2.0)**2*PAI
C
C*****
C
C      Calculation of Concentration Profile in Rock
C      (alternative condition)
C      (カンセキチュウノウト`フ`ンフ°ノケイサン(キョウカイノウト`カ`ヘンカスルハ`アイ))
C
C*****
C      Input of initial condition
C*****
C
DO 10 I=2,NZ+1
C(I,1)=CL
10 CONTINUE
C
C(1,1)=CH
C
C*****
C      Output of input data
C      (ニユウリョクシ`ョウケンノシュツリョク)
C*****
C

```

```
IF(P.EQ.0) THEN  
WRITE(6,1000) P  
ELSEIF(P.EQ.1) THEN  
WRITE(6,1001) P  
ELSE  
WRITE(6,1005)  
GO TO 500  
ENDIF
```

C

```
IF(NU.EQ.1) THEN  
WRITE(6,100) CH  
ELSEIF(NU.EQ.2) THEN  
WRITE(6,101) CH  
ELSEIF(NU.EQ.3) THEN  
WRITE(6,102) CH  
ELSEIF(NU.EQ.4) THEN  
WRITE(6,103) CH  
ENDIF
```

C

```
IF(NU.EQ.1) THEN  
WRITE(6,110) CL  
ELSEIF(NU.EQ.2) THEN  
WRITE(6,111) CL  
ELSEIF(NU.EQ.3) THEN  
WRITE(6,112) CL  
ELSEIF(NU.EQ.4) THEN  
WRITE(6,113) CL  
ENDIF
```

C

```
IF(NU.EQ.1) THEN  
WRITE(6,115) NU  
ELSEIF(NU.EQ.2) THEN  
WRITE(6,116) NU  
ELSEIF(NU.EQ.3) THEN  
WRITE(6,117) NU  
ELSEIF(NU.EQ.4) THEN  
WRITE(6,118) NU  
ENDIF
```

C

```
WRITE(6,120) DA  
WRITE(6,130) DE  
WRITE(6,140) Z  
WRITE(6,141) NZ  
WRITE(6,150) DT  
WRITE(6,160) D  
WRITE(6,165) VT  
WRITE(6,170) VM
```

C

```
IF(ND.EQ.1) THEN  
WRITE(6,171) ND  
ELSE  
WRITE(6,172) ND  
ENDIF
```

C

```
WRITE(6,176) NT  
WRITE(6,179) TC
```

```

C
  TCD=TC/24.0
  WRITE(6,182) TCD
C
  WRITE(6,181) NTP
C
C*****
C      Output to printer of input data
C      (ニューリヨクシヨウケンノフ° リンターヘノシュツリヨク)
C*****
C      Judgement of printing to printer
C      (フ° リンターヘノシュツリヨクハンテイ)
C*****
C
  IF(P.EQ.0) THEN
  GO TO 700
  ELSEIF(P.EQ.1) THEN
  WRITE(7,1001) P
  ENDIF
C
  IF(NU.EQ.1) THEN
  WRITE(7,100) CH
  ELSEIF(NU.EQ.2) THEN
  WRITE(7,101) CH
  ELSEIF(NU.EQ.3) THEN
  WRITE(7,102) CH
  ELSEIF(NU.EQ.4) THEN
  WRITE(7,103) CH
  ENDIF
C
  IF(NU.EQ.1) THEN
  WRITE(7,110) CL
  ELSEIF(NU.EQ.2) THEN
  WRITE(7,111) CL
  ELSEIF(NU.EQ.3) THEN
  WRITE(7,112) CL
  ELSEIF(NU.EQ.4) THEN
  WRITE(7,113) CL
  ENDIF
C
  IF(NU.EQ.1) THEN
  WRITE(7,115) NU
  ELSEIF(NU.EQ.2) THEN
  WRITE(7,116) NU
  ELSEIF(NU.EQ.3) THEN
  WRITE(7,117) NU
  ELSEIF(NU.EQ.4) THEN
  WRITE(7,118) NU
  ENDIF
C
  WRITE(7,120) DA
  WRITE(7,130) DE
  WRITE(7,140) Z
  WRITE(7,141) NZ
  WRITE(7,150) DT
  WRITE(7,160) D
  WRITE(7,165) VT

```



```

WRITE(7,170) VM
C
IF(ND.EQ.1) THEN
WRITE(7,171) ND
ELSE
WRITE(7,172) ND
ENDIF
C
WRITE(7,176) NT
WRITE(7,179) TC
WRITE(7,182) TCD
WRITE(7,181) NTP
C
C*****
C      Conversion of time unit: from hr to sec
C      (シゝカンタンイノヘンカン: hrカラsecへ)
C*****
C      700 DT=DT*3600.0
C
C*****
C      Judgement of dispersion
C      (ハツサンノハンテイ)
C*****
C      DD=DA*DT/DX**2
C
IF(DD.GE.0.5) THEN
WRITE(6,178) DD
GO TO 500
ENDIF
C
C*****
C      Calculation of flux
C      (フラックスノケイサン)
C*****
C      T=0.0
C      L=0
C      DO 400 N=1,1000
C      DO 20 J=1,250
C      FLAX1=DE*(C(1,J)-C(2,J))/DX
C      FLAX2=DE*(C(NZ,J)-C(NZ+1,J))/DX
C
C*****
C      Calculation of boundary condition
C      (キョウカイノウトノケイサン)
C*****
C      C(1,J+1)=C(1,J)-DT*FLAX1*A/VT
C      C(NZ+1,J+1)=C(NZ+1,J)+DT*FLAX2*A/VM
C
C*****
C      Calculation of concentration profile
C      (ノウトフンブノケイサン)
C*****

```

```

C
  DO 30 I=2,NZ
  C(I,J+1)=C(I,J)+DA*DT/DX**2
  &      *(C(I+1,J)-2.0*C(I,J)+C(I-1,J))
  30 CONTINUE
C
C*****
C
C      Calculation of Concentration Profile in Rock
C              (constant condition)
C      (カンセキチュウノアウト`ファンフ`ノケイサン(イツテキヨウカイノウト`ノハ`アイ))
C
C*****
C              Input of boundary condition
C              (キヨウカイシ`ヨウケンノニュウリョク)
C*****
C
  DO 40 K=1,251
  CC(1,K)=CH
  CC(NZ+1,K)=CL
  40 CONTINUE
C
C*****
C              Input of initial condition
C              (シヨクシ`ヨウケンノニュウリョク)
C*****
C
  IF(N.EQ.1) THEN
  DO 50 I=2,NZ
  CC(I,1)=CL
  50 CONTINUE
  ENDIF
C
C*****
C              Calculation of flux
C              (フラックスノケイサン)
C*****
C
  FFLAX1=DE*(CC(1,J)-CC(2,J))/DX
  FFLAX2=DE*(CC(NZ,J)-CC(NZ+1,J))/DX
C
C*****
C              Calculation of concentration profile
C              (ノウト`ファンフ`ノケイサン)
C*****
C
  DO 60 I=2,NZ
  CC(I,J+1)=CC(I,J)+DA*DT/DX**2
  &      *(CC(I+1,J)-2.0*CC(I,J)+CC(I-1,J))
  60 CONTINUE
C
C*****
C      Accumulation of time and conversion of unit: from
C      sec to hr and day
C      (シ`カンノセキサントタンイノヘンカン: secカラhrオヨヒ`day)
C*****

```

```

C
  IF(J.EQ.1.AND.N.EQ.1) THEN
  T=T
  ELSE
  T=T+DT/3600.0
  TD=T/24.0
  ENDIF

C
C*****
C
C          Output of Results
C          (ケツカノシュツリヨク)
C*****
C
  WRITE(6,180) T
  WRITE(6,185) TD

C
  IF(NU.EQ.1) THEN
  WRITE(6,190) FLAX1
  ELSEIF(NU.EQ.2) THEN
  WRITE(6,191) FLAX1
  ELSEIF(NU.EQ.3) THEN
  WRITE(6,192) FLAX1
  ELSEIF(NU.EQ.4) THEN
  WRITE(6,193) FLAX1
  ENDIF

C
  IF(NU.EQ.1) THEN
  WRITE(6,200) FLAX2
  ELSEIF(NU.EQ.2) THEN
  WRITE(6,201) FLAX2
  ELSEIF(NU.EQ.3) THEN
  WRITE(6,202) FLAX2
  ELSEIF(NU.EQ.4) THEN
  WRITE(6,203) FLAX2
  ENDIF

C
  IF(NU.EQ.1) THEN
  WRITE(6,210) (C(I,J),I=1,NZ+1)
  ELSEIF(NU.EQ.2) THEN
  WRITE(6,211) (C(I,J),I=1,NZ+1)
  ELSEIF(NU.EQ.3) THEN
  WRITE(6,212) (C(I,J),I=1,NZ+1)
  ELSEIF(NU.EQ.4) THEN
  WRITE(6,213) (C(I,J),I=1,NZ+1)
  ENDIF

C
  IF(NU.EQ.1) THEN
  WRITE(6,220) FFLAX1
  ELSEIF(NU.EQ.2) THEN
  WRITE(6,221) FFLAX1
  ELSEIF(NU.EQ.3) THEN
  WRITE(6,222) FFLAX1
  ELSEIF(NU.EQ.4) THEN
  WRITE(6,223) FFLAX1
  ENDIF

```

```

C
  IF(NU.EQ.1) THEN
  WRITE(6,230) FFLAX2
  ELSEIF(NU.EQ.2) THEN
  WRITE(6,231) FFLAX2
  ELSEIF(NU.EQ.3) THEN
  WRITE(6,232) FFLAX2
  ELSEIF(NU.EQ.4) THEN
  WRITE(6,233) FFLAX2
  ENDIF

C
  IF(NU.EQ.1) THEN
  WRITE(6,240) (CC(I,J),I=1,NZ+1)
  ELSEIF(NU.EQ.2) THEN
  WRITE(6,241) (CC(I,J),I=1,NZ+1)
  ELSEIF(NU.EQ.3) THEN
  WRITE(6,242) (CC(I,J),I=1,NZ+1)
  ELSEIF(NU.EQ.4) THEN
  WRITE(6,243) (CC(I,J),I=1,NZ+1)
  ENDIF

C
C*****
C      Output to printer of calculated results
C      (ケイサンケツカノフ° リンター-ヘノシュツリョク)
C*****
C      Judgement of printing to printer
C      (フ° リンター-ヘノシュツリョクハンテイ)
C*****
C
  IF(P.EQ.0) THEN
  GO TO 800
  ELSEIF(P.EQ.1) THEN
  GO TO 1300
  ENDIF

C
1300 IF(NTP.EQ.0) THEN
  GO TO 800
  ENDIF

C
  DO 600 K=1,NTP

C
  IF(T.EQ.TP(K)) THEN
  WRITE(7,180) T
  WRITE(7,185) TD

C
  IF(NU.EQ.1) THEN
  WRITE(7,190) FLAX1
  ELSEIF(NU.EQ.2) THEN
  WRITE(7,191) FLAX1
  ELSEIF(NU.EQ.3) THEN
  WRITE(7,192) FLAX1
  ELSEIF(NU.EQ.4) THEN
  WRITE(7,193) FLAX1
  ENDIF

C

```

```

        IF(NU.EQ.1) THEN
        WRITE(7,200) FLAX2
        ELSEIF(NU.EQ.2) THEN
        WRITE(7,201) FLAX2
        ELSEIF(NU.EQ.3) THEN
        WRITE(7,202) FLAX2
        ELSEIF(NU.EQ.4) THEN
        WRITE(7,203) FLAX2
        ENDIF
C
        IF(NU.EQ.1) THEN
        WRITE(7,210) (C(I,J),I=1,NZ+1)
        ELSEIF(NU.EQ.2) THEN
        WRITE(7,211) (C(I,J),I=1,NZ+1)
        ELSEIF(NU.EQ.3) THEN
        WRITE(7,212) (C(I,J),I=1,NZ+1)
        ELSEIF(NU.EQ.4) THEN
        WRITE(7,213) (C(I,J),I=1,NZ+1)
        ENDIF
C
        IF(NU.EQ.1) THEN
        WRITE(7,220) FFLAX1
        ELSEIF(NU.EQ.2) THEN
        WRITE(7,221) FFLAX1
        ELSEIF(NU.EQ.3) THEN
        WRITE(7,222) FFLAX1
        ELSEIF(NU.EQ.4) THEN
        WRITE(7,223) FFLAX1
        ENDIF
C
        IF(NU.EQ.1) THEN
        WRITE(7,230) FFLAX2
        ELSEIF(NU.EQ.2) THEN
        WRITE(7,231) FFLAX2
        ELSEIF(NU.EQ.3) THEN
        WRITE(7,232) FFLAX2
        ELSEIF(NU.EQ.4) THEN
        WRITE(7,233) FFLAX2
        ENDIF
C
        IF(NU.EQ.1) THEN
        WRITE(7,240) (CC(I,J),I=1,NZ+1)
        ELSEIF(NU.EQ.2) THEN
        WRITE(7,241) (CC(I,J),I=1,NZ+1)
        ELSEIF(NU.EQ.3) THEN
        WRITE(7,242) (CC(I,J),I=1,NZ+1)
        ELSEIF(NU.EQ.4) THEN
        WRITE(7,243) (CC(I,J),I=1,NZ+1)
        ENDIF
C
        ENDIF
C
        600 CONTINUE
C

```

```

C*****
C   Judgement of time whether to be calculated or not
C       (ケイサンシタイシ`カンカト`ウカノハンテイ)
C*****
C
C   800 IF(T.GE.TC) THEN
C         GO TO 500
C         ENDIF
C
C*****
C
C       Input of Information for Exchange of Tracer
C       (ヨウエキノコウカンシ`ヨウホウノニューリョク)
C*****
C
C       IF(ND.EQ.1) THEN
C
C*****
C       Input of time exchanged tracer in the tracer cell
C       (トレーサーセルチュウテ`ノトレーサーノコウカンシ`カンノニューリョク)
C*****
C
C         DO 80 I=1,NT
C           IF(T.EQ.TT(I)) THEN
C             L=L+1
C             WRITE(6,250) L
C             C(1,J+1)=CHC(L)
C             ENDIF
C         80 CONTINUE
C
C       ENDIF
C
C     20 CONTINUE
C
C*****
C       Description of format
C       (フォーマットノキシ`ユツ)
C*****
C
C   100 FORMAT (2X,'シヨキノコウノウト`カ`ワノウト` Ch=',F10.2,1X,
C         &      'Bq/ml')
C   101 FORMAT (2X,'シヨキノコウノウト`カ`ワノウト` Ch=',F10.2,1X,
C         &      'ppm')
C   102 FORMAT (2X,'シヨキノコウノウト`カ`ワノウト` Ch=',F10.2,1X,
C         &      'ppb')
C   103 FORMAT (2X,'シヨキノコウノウト`カ`ワノウト` Ch=',F10.2,1X,
C         &      'M')
C
C   110 FORMAT (2X,'シヨキノテイノウト`カ`ワノウト` C1=',F10.2,1X,
C         &      'Bq/ml')
C   111 FORMAT (2X,'シヨキノテイノウト`カ`ワノウト` C1=',F10.2,1X,
C         &      'ppm')
C   112 FORMAT (2X,'シヨキノテイノウト`カ`ワノウト` C1=',F10.2,1X,
C         &      'ppb')
C   113 FORMAT (2X,'シヨキノテイノウト`カ`ワノウト` C1=',F10.2,1X,
C         &      'M')

```

- C
- 115 FORMAT (2X, 'ノウト`ノタンイハンテイキコ`ウ NU=', 1I1, 1X,
& 'タンイ:Bq/ml')
- 116 FORMAT (2X, 'ノウト`ノタンイハンテイキコ`ウ NU=', 1I1, 1X,
& 'タンイ:ppm')
- 117 FORMAT (2X, 'ノウト`ノタンイハンテイキコ`ウ NU=', 1I1, 1X,
& 'タンイ:ppb')
- 118 FORMAT (2X, 'ノウト`ノタンイハンテイキコ`ウ NU=', 1I1, 1X,
& 'タンイ:M')
- C
- 120 FORMAT (2X, 'ミカケノカクサンケイスウ Da=', G20.3, 1X,
& 'cm2/s')
- 130 FORMAT (2X, 'シ`ッコウカクサンケイスウ De=', G20.3, 1X,
& 'cm2/s')
- 140 FORMAT (2X, 'カンセキノアツサ Z=', F10.2, 1X, 'cm')
- 141 FORMAT (2X, 'アツサホウコウノフ`ンカツスウ Nz=', 1I2, 1X, 'フ`ンカツ')
- 150 FORMAT (2X, 'シ`カンノキサ`ミ DT=', F10.2, 1X, 'hr')
- 160 FORMAT (2X, 'サンポ`ルノチヨツケイ D=', F10.2, 1X, 'cm')
- 165 FORMAT (2X, 'トレ`サーセルチュウノヨウエキノタイセキ Vt=', F10.1, 1X,
& 'ml')
- 170 FORMAT (2X, 'ソクテイセルチュウノヨウエキノタイセキ Vm=', F10.1, 1X,
& 'ml')
- 171 FORMAT (2X, 'トレ`サーノコウカンノハンテイスウ ND=', 1I1, 1X,
& 'コウカンシル')
- 172 FORMAT (2X, 'トレ`サーノコウカンノハンテイスウ ND=', 1I1, 1X,
& 'コウカンシナイ')
- 176 FORMAT (2X, 'トレ`サーノコウカンカイスウ NT=', 1I2, 1X, 'カイ')
- 177 FORMAT (2X, 'NTカ`1ト20ノアイタ`テ`ナケレハ`イケマセン')
- 178 FORMAT (2X, 'ハツサンシマス Dd=', F7.2)
- 179 FORMAT (2X, 'ケイサンシタイシ`カン Tc=', F10.2, 1X, 'hr')
- 180 FORMAT (/ , 2X, 'ケイカシ`カン T=', F10.2, 1X, 'hr')
- 181 FORMAT (2X, 'ケイサンケツカヲフ`リントアウトスルカイスウ NTP=', 1I2,
& 1X, 'カイ')
- 182 FORMAT (2X, 'ケイサンシタイニツスウ Tcd=', F10.2, 1X, 'day')
- 185 FORMAT (2X, 'ケイカニツスウ Td=', F10.2, 1X, 'day'/)
- C
- 190 FORMAT (2X, 'X=0テ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX1=',
& , G20.3, 1X, 'Bq/cm2/s')
- 191 FORMAT (2X, 'X=0テ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX1=',
& , G20.3, 1X, 'ug/cm2/s')
- 192 FORMAT (2X, 'X=0テ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX1=',
& , G20.3, 1X, 'ng/cm2/s')
- 193 FORMAT (2X, 'X=0テ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX1=',
& , G20.3, 1X, 'mM/cm2/s')
- C
- 200 FORMAT (2X, 'X=Zテ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX2=',
& , G20.3, 1X, 'Bq/cm2/s'/)
- 201 FORMAT (2X, 'X=Zテ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX2=',
& , G20.3, 1X, 'ug/cm2/s'/)
- 202 FORMAT (2X, 'X=Zテ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX2=',
& , G20.3, 1X, 'ng/cm2/s'/)
- 203 FORMAT (2X, 'X=Zテ`ノフラックス(キョウカイシ`ヨウケンヘンカ) FLAX2=',
& , G20.3, 1X, 'mM/cm2/s'/)
- C
- 210 FORMAT (2X, 'ノウト` (キョウカイシ`ヨウケンヘンカ) C=', 5F10.2,
& 1X, 'Bq/ml')

```

211 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンヘンカ) C= ', 5F10.2,
& 1X, 'ppm')
212 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンヘンカ) C= ', 5F10.2,
& 1X, 'ppb')
213 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンヘンカ) C= ', 5F10.2,
& 1X, 'M')
C
220 FORMAT (/ , 2X, 'X=0テ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX1= '
& , G20.3, 1X, 'Bq/cm2/s ')
221 FORMAT (/ , 2X, 'X=0テ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX1= '
& , G20.3, 1X, 'ug/cm2/s ')
222 FORMAT (/ , 2X, 'X=0テ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX1= '
& , G20.3, 1X, 'ng/cm2/s ')
223 FORMAT (/ , 2X, 'X=0テ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX1= '
& , G20.3, 1X, 'mM/cm2/s ')
C
230 FORMAT (2X, 'X=Zテ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX2= '
& , G20.3, 1X, 'Bq/cm2/s '/')
231 FORMAT (2X, 'X=Zテ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX2= '
& , G20.3, 1X, 'ug/cm2/s '/')
232 FORMAT (2X, 'X=Zテ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX2= '
& , G20.3, 1X, 'ng/cm2/s '/')
233 FORMAT (2X, 'X=Zテ` ノフラックス(キョウカイシ` ヨウケンイツテイ) FFLAX2= '
& , G20.3, 1X, 'mM/cm2/s '/')
C
240 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンイツテイ) CC= ', 5F10.2,
& 1X, 'Bq/ml ')
241 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンイツテイ) CC= ', 5F10.2,
& 1X, 'ppm')
242 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンイツテイ) CC= ', 5F10.2,
& 1X, 'ppb')
243 FORMAT (2X, 'ノウト` (キョウカイシ` ヨウケンイツテイ) CC= ', 5F10.2,
& 1X, 'M')
C
250 FORMAT (/ , 2X, 'L= ', 1I5)
C
1000 FORMAT (2X, 'フ` リンタ-ヘノシュツリヨクハンテイ P= ', 1I1, 1X, 'シナイ')
1001 FORMAT (2X, 'フ` リンタ-ヘノシュツリヨクハンテイ P= ', 1I1, 1X, 'スル')
C
1005 FORMAT (2X, 'フ` リンタ-ヘノシュツリヨクハンテイ スウ、P=1:スル、P=0:シナイテ` ス')
C
1010 FORMAT (2X, 'アツサホウコウノフ` ンカツスウハ、5, 10, 15, 20ノト` レカテ` ス')
C
C*****
C
DO 300 I=1, NZ+1
C(I, 1)=C(I, 251)
300 CONTINUE
C
DO 310 I=1, NZ+1
CC(I, 1)=CC(I, 251)
310 CONTINUE
C
400 CONTINUE

```



```
C
C*****
C
C   Declaration of End of Program and Close of File
C   (プログラムノシユウリヨウオヨヒ`ファイルノハイソクセンゲ`ン)
C
C*****
C
C   CLOSE(1)
C   CLOSE(7)
500 STOP
END
```